```
(* We compute the flux of F through the paraboloid z=
 x^2+y^2 below the plane z=1 oriented down *)
```

In[1]:=
```
F = {x, y, z^4}
```

Out[1]= $\{x, y, z^4\}$

In[2]:=
```
(* parametrization using cylindrical coordinates*)
q = {r Cos[t], r Sin[t], r^2}
qr = D[q, r]
qt = D[q, t]
```

Out[2]= $\{r\,\mathrm{Cos}[t], r\,\mathrm{Sin}[t], r^2\}$

Out[3]= $\{\mathrm{Cos}[t], \mathrm{Sin}[t], 2\,r\}$

Out[4]= $\{-r\,\mathrm{Sin}[t], r\,\mathrm{Cos}[t], 0\}$

In[5]:=
```
(* normal vector *)
n = Simplify[Cross[qt, qr]]
```

Out[5]= $\{2\,r^2\,\mathrm{Cos}[t], 2\,r^2\,\mathrm{Sin}[t], -r\}$

In[6]:=
```
(* Double integral *)
Fcompq = F /. {x → q[[1]], y → q[[2]], z → q[[3]]}
g = Dot[Fcompq, n]
Integrate[Integrate[g, {r, 0, 1 }], {t, 0, 2 Pi}]
```

Out[6]= $\{r\,\mathrm{Cos}[t], r\,\mathrm{Sin}[t], r^8\}$

Out[7]= $-r^9 + 2\,r^3\,\mathrm{Cos}[t]^2 + 2\,r^3\,\mathrm{Sin}[t]^2$

Out[8]= $\frac{4\,\pi}{5}$

In[14]:=
```
(* alternate simpler parametrization, followed by change of variable *)
q = {x, y , x^2 + y^2}
qx = D[q, x]
qy = D[q, y]
```

Out[14]= $\{x, y, x^2 + y^2\}$

Out[15]= $\{1, 0, 2\,x\}$

Out[16]= $\{0, 1, 2\,y\}$

In[17]:=
```
(* Normal vector *)
n = Simplify[Cross[qy, qx]]
```

Out[17]= $\{2\,x, 2\,y, -1\}$

In[18]:=
```
Fcompq = F /. {x → q[[1]], y → q[[2]], z → q[[3]]}
g = Dot[Fcompq, n]
```

Out[18]= $\{x, y, (x^2 + y^2)^4\}$

Out[19]= $2 x^2 + 2 y^2 - (x^2 + y^2)^4$

In[22]:=
```
(* switch to polar coordinates to handle the region of integration (disk) *)
gg = Simplify[g /. {x → r Cos[t], y → r Sin[t]}]
Integrate[Integrate[gg r, {r, 0, 1 }], {t, 0, 2 Pi}]
```

Out[22]= $-r^2 (-2 + r^6)$

Out[23]= $\frac{4 \pi}{5}$